

# NAG Library Routine Document

## F04BGF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F04BGF computes the solution to a real system of linear equations  $AX = B$ , where  $A$  is an  $n$  by  $n$  symmetric positive definite tridiagonal matrix and  $X$  and  $B$  are  $n$  by  $r$  matrices. An estimate of the condition number of  $A$  and an error bound for the computed solution are also returned.

### 2 Specification

```
SUBROUTINE F04BGF (N, NRHS, D, E, B, LDB, RCOND, ERBND, IFAIL)
  INTEGER          N, NRHS, LDB, IFAIL
  REAL (KIND=nag_wp) D(*), E(*), B(LDB,*), RCOND, ERBND
```

### 3 Description

$A$  is factorized as  $A = LDL^T$ , where  $L$  is a unit lower bidiagonal matrix and  $D$  is diagonal, and the factored form of  $A$  is then used to solve the system of equations.

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

### 5 Arguments

- 1: N – INTEGER *Input*  
*On entry:* the number of linear equations  $n$ , i.e., the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 2: NRHS – INTEGER *Input*  
*On entry:* the number of right-hand sides  $r$ , i.e., the number of columns of the matrix  $B$ .  
*Constraint:*  $NRHS \geq 0$ .
- 3: D(\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array D must be at least  $\max(1, N)$ .  
*On entry:* must contain the  $n$  diagonal elements of the tridiagonal matrix  $A$ .  
*On exit:* if  $IFAIL = 0$  or  $N + 1$ , D is overwritten by the  $n$  diagonal elements of the diagonal matrix  $D$  from the  $LDL^T$  factorization of  $A$ .
- 4: E(\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array E must be at least  $\max(1, N - 1)$ .  
*On entry:* must contain the  $(n - 1)$  subdiagonal elements of the tridiagonal matrix  $A$ .

*On exit:* if IFAIL = 0 or N + 1, E is overwritten by the  $(n - 1)$  subdiagonal elements of the unit lower bidiagonal matrix  $L$  from the  $LDL^T$  factorization of  $A$ . (E can also be regarded as the superdiagonal of the unit upper bidiagonal factor  $U$  from the  $U^T DU$  factorization of  $A$ .)

5: B(LDB, \*) – REAL (KIND=nag\_wp) array *Input/Output*

**Note:** the second dimension of the array B must be at least  $\max(1, \text{NRHS})$ .

*On entry:* the  $n$  by  $r$  matrix of right-hand sides  $B$ .

*On exit:* if IFAIL = 0 or N + 1, the  $n$  by  $r$  solution matrix  $X$ .

6: LDB – INTEGER *Input*

*On entry:* the first dimension of the array B as declared in the (sub)program from which F04BGF is called.

*Constraint:*  $LDB \geq \max(1, N)$ .

7: RCOND – REAL (KIND=nag\_wp) *Output*

*On exit:* if IFAIL = 0 or N + 1, an estimate of the reciprocal of the condition number of the matrix  $A$ , computed as  $RCOND = 1 / (\|A\|_1 \|A^{-1}\|_1)$ .

8: ERRBND – REAL (KIND=nag\_wp) *Output*

*On exit:* if IFAIL = 0 or N + 1, an estimate of the forward error bound for a computed solution  $\hat{x}$ , such that  $\|\hat{x} - x\|_1 / \|x\|_1 \leq \text{ERRBND}$ , where  $\hat{x}$  is a column of the computed solution returned in the array B and  $x$  is the corresponding column of the exact solution  $X$ . If RCOND is less than **machine precision**, then ERRBND is returned as unity.

9: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL > 0 and IFAIL ≤ N

The principal minor of order  $\langle \text{value} \rangle$  of the matrix  $A$  is not positive definite. The factorization has not been completed and the solution could not be computed.

IFAIL = N + 1

A solution has been computed, but RCOND is less than **machine precision** so that the matrix  $A$  is numerically singular.

IFAIL = -1

On entry, N =  $\langle value \rangle$ .  
Constraint:  $N \geq 0$ .

IFAIL = -2

On entry, NRHS =  $\langle value \rangle$ .  
Constraint:  $NRHS \geq 0$ .

IFAIL = -6

On entry, LDB =  $\langle value \rangle$  and N =  $\langle value \rangle$ .  
Constraint:  $LDB \geq \max(1, N)$ .

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

*The real allocatable memory required is N. In this case the factorization and the solution X have been computed, but RCOND and ERRBND have not been computed.*

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

The computed solution for a single right-hand side,  $\hat{x}$ , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and  $\epsilon$  is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where  $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$ , the condition number of  $A$  with respect to the solution of the linear equations. F04BGF uses the approximation  $\|E\|_1 = \epsilon \|A\|_1$  to estimate ERRBND. See Section 4.4 of Anderson *et al.* (1999) for further details.

## 8 Parallelism and Performance

F04BGF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations required to solve the equations  $AX = B$  is proportional to  $nr$ . The condition number estimation requires  $O(n)$  floating-point operations.

See Section 15.3 of Higham (2002) for further details on computing the condition number of tridiagonal matrices.

The complex analogue of F04BGF is F04CGF.

## 10 Example

This example solves the equations

$$AX = B,$$

where  $A$  is the symmetric positive definite tridiagonal matrix

$$A = \begin{pmatrix} 4.0 & -2.0 & 0 & 0 & 0 \\ -2.0 & 10.0 & -6.0 & 0 & 0 \\ 0 & -6.0 & 29.0 & 15.0 & 0 \\ 0 & 0 & 15.0 & 25.0 & 8.0 \\ 0 & 0 & 0 & 8.0 & 5.0 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 6.0 & 10.0 \\ 9.0 & 4.0 \\ 2.0 & 9.0 \\ 14.0 & 65.0 \\ 7.0 & 23.0 \end{pmatrix}.$$

An estimate of the condition number of  $A$  and an approximate error bound for the computed solutions are also printed.

### 10.1 Program Text

```

Program f04bgfe

!      F04BGF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: f04bgf, nag_wp, x04caf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: errbnd, rcond
Integer                     :: i, ierr, ifail, ldb, n, nrhs
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: b(:,:), d(:), e(:)
!      .. Executable Statements ..
Write (nout,*) 'F04BGF Example Program Results'
Write (nout,*)
Flush (nout)
!      Skip heading in data file
Read (nin,*)
Read (nin,*) n, nrhs
ldb = n
Allocate (b(ldb,nrhs),d(n),e(n-1))
!      Read A from data file
Read (nin,*) d(1:n)
Read (nin,*) e(1:n-1)

!      Read B from data file
Read (nin,*)(b(i,1:nrhs),i=1,n)

!      Solve the equations AX = B for X

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 1
Call f04bgf(n,nrhs,d,e,b,ldb,rcond,errbnd,ifail)

```

```

      If (ifail==0) Then
!       Print solution, estimate of condition number and approximate
!       error bound

      ierr = 0
      Call x04caf('General',' ',n,nrhs,b,ldb,'Solution',ierr)

      Write (nout,*)
      Write (nout,*) 'Estimate of condition number'
      Write (nout,99999) 1.0E0_nag_wp/rcond
      Write (nout,*)
      Write (nout,*) 'Estimate of error bound for computed solutions'
      Write (nout,99999) errbnd
    Else If (ifail==n+1) Then
!       Matrix A is numerically singular. Print estimate of
!       reciprocal of condition number and solution
      Write (nout,*)
      Write (nout,*) 'Estimate of reciprocal of condition number'
      Write (nout,99999) rcond
      Write (nout,*)
      Flush (nout)

      ierr = 0
      Call x04caf('General',' ',n,nrhs,b,ldb,'Solution',ierr)

    Else If (ifail>0 .And. ifail<=n) Then
      Write (nout,99998) 'The leading minor of order ', ifail,
        ' is not positive definite'
    Else
      Write (nout,99997) ifail
    End If

99999 Format (6X,1P,E9.1)
99998 Format (1X,A,I3,A)
99997 Format (1X,' ** F04BGF returned with IFAIL = ',I5)
      End Program f04bgfe

```

## 10.2 Program Data

F04BGF Example Program Data

```

  5      2                :Values of N and NRHS

  4.0  10.0  29.0  25.0   5.0 :End of diagonal D
-2.0  -6.0  15.0   8.0      :End of sub-diagonal E

  6.0  10.0
  9.0   4.0
  2.0   9.0
14.0  65.0
  7.0  23.0                :End of matrix B

```

## 10.3 Program Results

F04BGF Example Program Results

```

Solution
      1      2
1      2.5000   2.0000
2      2.0000  -1.0000
3      1.0000  -3.0000
4     -1.0000   6.0000
5      3.0000  -5.0000

```

Estimate of condition number  
1.1E+02

Estimate of error bound for computed solutions  
1.2E-14

---